Thanks to the hint to the correct function by Joafigue, I solved the question by adding this to my config:

;; Recalculate all org tables in the buffer when saving.
(defvar-local modi/org-table-enable-buffer-wide-recalculation t
"When non-nil, all the org tables in the buffer will be recalculated when
saving the file.
This variable is buffer local.")
;; Mark `modi/org-table-enable-buffer-wide-recalculation' as a safe local
;; variable as long as its value is t or nil. That way you are not prompted
;; to add that to `safe-local-variable-values' in custom.el.
(put 'modi/org-table-enable-buffer-wide-recalculation 'safe-local-variable #'booleanp)
(defun modi/org-table-recalculate-buffer-tables (&rest args)
"Wrapper function for `org-table-recalculate-buffer-tables' that runs
that function only if `modi/org-table-enable-buffer-wide-recalculation' is
non-nil.
Also, this function has optional ARGS that is needed for any function that is
added to `org-export-before-processing-hook'. This would be useful if this
function is ever added to that hook."
(when modi/org-table-enable-buffer-wide-recalculation
(org-table-recalculate-buffer-tables)))
(defun modi/org-table-recalculate-before-save ()
"Recalculate all org tables in the buffer before saving."
(add-hook 'before-save-hook #'modi/org-table-recalculate-buffer-tables nil :local))
(add-hook 'org-mode-hook #'modi/org-table-recalculate-before-save)